# Exame Final Vibrações Questão 01

## 1º Semestre de 2015

### 29/07/2015

Como o sistema tem inércia rotativa, o primeiro passo é calcular a massa equivalente do cilindro.

Pela igualdade das energias cinéticas:

```
var('m, m_eq, J0, d, theta, omega, v, v_eq')
eq= 1/2*m_eq*v_eq^2 == 1/2*m*v^2 + 1/2*J0*omega^2
show(eq)
```

$$\frac{1}{2}\,m_{eq}v_{eq}^2 = \frac{1}{2}\,J_0\omega^2 + \frac{1}{2}\,mv^2$$

Substituindo a velocidade e a velocidade angular

```
eq2=eq.substitute({omega:2*v_eq/d, v:v_eq})
show(eq2)
```

$$\frac{1}{2}\,m_{eq}v_{eq}^2 = \frac{1}{2}\,mv_{eq}^2 + \frac{2\,J_0v_{eq}^2}{d^2}$$

Substituindo o momento de inércia

```
eq3=eq2.subs({J0: m*d^2/8})
show(eq3)
```

$$\frac{1}{2}\,m_{eq}v_{eq}^2 = \frac{3}{4}\,mv_{eq}^2$$

Resolvendo para a massa equivalente

```
s=solve(eq3,m_eq)
show(s)
```

$$\left[m_{eq} = \frac{3}{2}\,m\right]$$

O que mostra que basta adicionar a metade da massa do cilindro para obtermos a massa equivalente. Este resultado é útil para termos na cabeça.

No caso temos:

```
d=0.1
h=0.008
A=N(pi*d^2/4)
V=A*h
rho=7700
m = rho*V
show([m, V, A])
```

$$[0.483805268652828, 0.0000628318530717959, 0.00785398163397448]$$

```
m_eq=1.5*m
show(m_eq)
```

$$0.725707902979242$$

A frequência natural é então

```
k=2000
omega_n=sqrt(k/m_eq)
show(omega_n)
```

$$52.4969499162476$$

Do enunciado, a amplitude e o período da força aplicada são

```
A=100
T=0.30
omega_0=N(2*pi)/T
show(omega_0)
```

$$20.9439510239320$$

A expressão para a força aplicada é então

```
var('n, t')
A02=A/2
An(n)=(4*A)/N(pi)*sin((2*n-1)*omega_0*t)/(2*n-1)
show(A02)
show(An)
```

$$50$$

$$n \mapsto \frac{127.323954473516 \sin((41.8879020478639\, n - 20.9439510239320)t)}{2\, n - 1}$$

Os dez primeiros termos são então (na prova não é, obviamente, para fazer com dez, só estou fazendo assim aqui porque é fácil no computador).

```
s=[An(j+1) for j in range(10)]
show(s)
```

$$[127.323954473516\ \sin(20.9439510239320\, t), 42.4413181578388\ \sin(62.83$$

Como não há amortecimento, os ângulos de fase são nulos e os coeficientes da resposta são dados por:

```
var('bj, r')
Bj(j)=(bj/k)/sqrt(1-(j*r)^2)
show(Bj)
Bj=Bj.subs({r:omega_0/omega_n})
show(Bj)
```

$$j \mapsto \frac{bj}{2000\sqrt{-j^2r^2+1}}$$

$$j \mapsto \frac{bj}{2000\sqrt{-0.159165553625539\, j^2+1}}$$

Calculando os dez primeiros termos (mesma coisa...)

```
bjs=[Bj(j+1) for j in range(10)]
show(bjs)
```

$$\Big[0.000545273958366573\, bj, 0.000829496813242033\, bj, \Big(4.65546251377357$$

Vamos separar a série da força nas amplitudes (que serão os $b_j$) e nas frequências, que serão as frequências das respostas. Isto é meio sofisticado, não espero que ninguém entenda o programa :)

```
amp=SR.wild(0)
freq=SR.wild(1)
pattern=amp*sin(freq*t)
tf=[s[j].match(pattern) for j in range(10)]; show(tf)
```

$$[\{\$1 : 20.9439510239320, \$0 : 127.323954473516\}, \{\$1 : 62.831853071795$$

Os primeiros termos em $t$ da resposta são então (sem o termo constante que eu vou colocar só no fim)

```
terms=[ tf[j][amp]*sin(tf[j][freq]*t) for j in range(10)]
show(terms)
```

$$[127.323954473516\ \sin(20.9439510239320\,t), 42.4413181578388\ \sin(62.83$$

A resposta total é então

```
allTerms = [ N(A02/k) ] + terms
show(allTerms)
x(t) = sum(allTerms)
show(x)
```

$$[0.0250000000000000, 127.323954473516\ \sin(20.943951023932$$

$$t \mapsto 6.70126076176401\ \sin(397.935069454707\,t) + 7.48964438079507\ \sin($$

(a constante 50 ficou lá no final da lista, mas está lá.)

Só por curiosidade.

```
plot(x,t,0, 3)
```



100
50
-50
-100
0.5
1
1.5
2

# Exame Final Vibrações Questão 02

## 1º Semestre 2015

### 29/07/2015

Devido ao enunciado podemos considerar as câmaras de gás como molas elásticas, que chamaremos de $k_1$, $k_2$ e $k_3$, da esquerda para a direita.

Assim, este problema torna-se exatamente o problema padrão com dois graus de liberdade, e podemos usar diretamente as fórmulas dadas. Como não há amortecimento o problema é realmente trivial.

As matrizes de massa e rigidez são

```
M=matrix([[1, 0],[0, 1]])
show(M)
var('k1, k2, k3, k')
K=matrix([[k1+k2, -k2],[-k2, k2+k3]])
show(K)
K=K.subs({k1:k, k2:k, k3:k})
show(K)
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} k_1 + k_2 & -k_2 \\ -k_2 & k_2 + k_3 \end{pmatrix}$$

$$\begin{pmatrix} 2\,k & -k \\ -k & 2\,k \end{pmatrix}$$

A matriz de impedância mecânica é

```
var('omega')
Ziw = K - omega^2*M
show(Ziw)
```

$$\begin{pmatrix} -\omega^2 + 2\,k & -k \\ -k & -\omega^2 + 2\,k \end{pmatrix}$$

No caso, temos *duas* frequências de excitação **diferentes,** então  temos que ter duas matrizes de impendância mecânica diferentes, uma para cada frequência.

```
Z15=Ziw.subs(omega=15.0)
Z25=Ziw.subs(omega=25.0)
show([Z15, Z25])
```

$$\left[\begin{pmatrix} 2\,k - 225.000000000000 & -k \\ -k & 2\,k - 225.000000000000 \end{pmatrix}, \left( \begin{matrix} 2\,k - 625.00 \end{matrix}\right.\right.$$

Introduzindo a rigidez

```
Z15=Z15.subs(k=480.0)
Z25=Z25.subs(k=480.0)
show([Z15, Z25])
```

$$\left[\begin{pmatrix} 735.000000000000 & -480.000000000000 \\ -480.000000000000 & 735.000000000000 \end{pmatrix}, \left( \begin{matrix} 335.000000000000 \\ -480.000000000000 \end{matrix}\right.\right.$$

Os vetores de forças externas são

```
F15 = vector([ 2.0, 0])
F25 = vector([ 0, 4.0])
```

As inversas das matrizes de impedância mecânica são

```
Z15i=Z15.inverse()
Z25i=Z25.inverse()
show([Z15i, Z25i])
```

$$\left[\begin{pmatrix} 0.00237230694747035 & 0.00154926167998063 \\ 0.00154926167998063 & 0.00237230694747035 \end{pmatrix}, \left( \begin{matrix} -0.00283477892 \\ -0.00406177279 \end{matrix}\right.\right.$$

As respostas devidas às duas forças são então

```
X15=Z15i*F15
X25=Z25i*F25
show([X15, X25])
```

$$[(0.00474461389494069,\ 0.00309852335996127), (-0.0162470911783372,$$

A resposta total é então

```
var('t')
```

```
X=X15*cos(15*t)+X25*cos(25*t)
show(X)
x0=X[0]
x1=X[1]
show(x0)
show(x1)
plot([x0, x1], t, 0, 2)
```

$$(-0.0162470911783372\,\cos(25\,t) + 0.00474461389494069\,\cos(15\,t),\ -0.01$$

$$-0.0162470911783372\,\cos(25\,t) + 0.00$$

$$-0.0113391157182145\,\cos(25\,t) + 0.00$$




Claramente, estes foram 5 pontos dados de graça....

# Exame Final Vibrações Questão 03

## Primeiro Semestre 2015

### 29/07/2015

Este é claramente um problema de transmissibilidade de deslocamento, de solução trivial e imediata.

A frequência natural da motocicleta é

```
m=300.0
k=45000.0
omega_n=sqrt(k/m)
show(omega_n)
```

$$12.2474487139159$$

O comprimento de onda da ondulação é

```
l=35
```

O período da excitação causada pela estrada é

```
v=60.0*1000/3600
T=l/v
show(T)
```

$$2.10000000000000$$

A frequência e a frequência circular de excitação são

```
f=1/T
omega=N(2*pi*f)
show([omega, f])
```

$$[2.99199300341885, 0.476190476190476]$$

A transmissibilidade de deslocamente é

```
var('zeta, r')
Td=sqrt((1+(2*zeta*r)^2)/((1-r^2)^2+(2*zeta*r)^2))
show(Td)
```

$$\sqrt{\frac{4\,r^{2}\zeta^{2}+1}{4\,r^{2}\zeta^{2}+\left(r^{2}-1\right)^{2}}}$$

No caso,

```
r=omega/omega_n
zeta=0.15
T1=Td(zeta=zeta, r=r)
show(T1)
```

$$1.06309607531977$$

O deslocamento é então:

```
A=T1*0.2
show(A)
```

$$0.212619215063955$$

A pior velocidade para o deslocamento ocorre quando a amplitude de deslocamento é máxima, o que não ocorre na ressonância. Na ressonância não é nada bom também, neste caso, o deslocamento seria

```
Tr=Td(zeta=zeta, r=1)
A=Tr*0.2
show(Tr)
```

$$3.48010216963685$$

A ressonância ocorre para

```
fn=omega_n/N(2*pi)
vn=l*fn
show([vn, fn])
show(vn*3600/1000)
```

$$\left[68.2234701079467, 1.94924200308419\right]$$

$$245.604492388608$$

O que me parece pouco preocupante com uma motocicleta de rua normal, mas perfeitamente viável com uma motocicleta esportiva, que normalmente passam de

300km/h

A máxima amplitude occore realmente ocorre em

```
var('r')
d=Td.diff(r)
s=solve(d==0.0, r)
show(s)
rmax=s[1].rhs()
show(rmax)
```

$$\left[r = -\frac{\sqrt{\sqrt{8\,\zeta^2+1}-1}}{2\,\zeta}, r = \frac{\sqrt{\sqrt{8\,\zeta^2+1}-1}}{2\,\zeta}, r = -\frac{\sqrt{-\sqrt{8\,\zeta^2+1}-}}{2\,\zeta}\right.$$

$$\frac{\sqrt{\sqrt{8\,\zeta^2+1}-1}}{2\,\zeta}$$

No caso

```
rm=rmax(zeta=0.15)
show(rm)
```

$$0.979104177410382$$

Percebam que a diferença entre este valor e o $r$ da ressonância ($r = 1$) é desprezível na prática, então quem tomou o valor da ressonância como o pior valor está correto também!

A frequência de excitação é

```
omega_m = rm*omega_n
fm = omega_m/N(2*pi)
vm = l*fm
show([vm, omega_m, fm])
show(vm*3600/1000)
```

$$[66.7978845801229, 11.9915281984145, 1.90851098800351]$$

$$240.472384488442$$

Que, novamente, é muito rápido, mas é concebível para motos esportivas.